Конфигурация БП 3.0 . Займы ведутся на счете 58.03

Мне нужно выбрать остатки по дня за месяц, чтобы посчитать проценты по выданным займам. Но почему-то результат запроса пустой. Подскажите, где у меня ошибка

```1c
    Запрос= Новый Запрос;
    Запрос.УстановитьПараметр("НачалоМесяца",НачалоМесяца);
    Запрос.УстановитьПараметр("КонецМесяца",КонецМесяца);
    Запрос.УстановитьПараметр("Займы",ПланыСчетов.Хозрасчетный.ПредоставленныеЗаймы);
    Запрос.УстановитьПараметр("Контрагент",Контрагент);
    Запрос.УстановитьПараметр("Договор",Договор);
    ВидыСубконто = Новый Массив;
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры);
    Запрос.УстановитьПараметр("ВидыСубконто", ВидыСубконто);

    Запрос.Текст="ВЫБРАТЬ
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК СуммаНачальныйОстатокДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК СуммаОборотДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК СуммаОборотКт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК СуммаКонечныйОстатокДт,
    | ХозрасчетныйОстаткиИОбороты.Период КАК Период
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, , Счет = &Займы,&ВидыСубконто, Субконто1 = &Договор) КАК ХозрасчетныйОстаткиИОбороты
    |
    |СГРУППИРОВАТЬ ПО
    | ХозрасчетныйОстаткиИОбороты.Период";

    ТЗ=Запрос.Выполнить().Выгрузить();
```
---
Разобрался. Просто по контрагенту забито два разных договора ( с одинаковым наименованием) по разными типами. И остатки по договору с типом Прочее, а я пробую отобрать с типом "С покупателем".
---
Субконто1 = &Договор может Субконто1 = &Контрагент ?
---
Попробовал даже так

```1c
    Запрос= Новый Запрос;
    Запрос.УстановитьПараметр("НачалоМесяца",НачалоМесяца);
    Запрос.УстановитьПараметр("КонецМесяца",КонецМесяца);
    Запрос.УстановитьПараметр("Займы",ПланыСчетов.Хозрасчетный.ПредоставленныеЗаймы);
    Запрос.УстановитьПараметр("Контрагент",Контрагент);
    Запрос.УстановитьПараметр("Договор",Договор);
    Запрос.УстановитьПараметр("Организация",Организация);
    ВидыСубконто = Новый Массив;
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты);
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры);
    Запрос.УстановитьПараметр("ВидыСубконто", ВидыСубконто);

    Запрос.Текст="ВЫБРАТЬ
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК СуммаНачальныйОстатокДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК СуммаОборотДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК СуммаОборотКт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК СуммаКонечныйОстатокДт,
    | ХозрасчетныйОстаткиИОбороты.Период КАК Период,
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Субконто1,
    | ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Субконто2
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, , Счет = &Займы, &ВидыСубконто, Организация = &Организация) КАК ХозрасчетныйОстаткиИОбороты
    |ГДЕ
    | ХозрасчетныйОстаткиИОбороты.Субконто1 = &Контрагент
    | И ХозрасчетныйОстаткиИОбороты.Субконто2 = &Договор
    |
    |СГРУППИРОВАТЬ ПО
    | ХозрасчетныйОстаткиИОбороты.Период,
    | ХозрасчетныйОстаткиИОбороты.Субконто1,
    | ХозрасчетныйОстаткиИОбороты.Субконто2";
```

Но результат тот же - Пустой
---
Мне что нужно два отдельных запроса, сначала по остаткам, а затем по оборотам?
---
попробуйте сначала просто в консоли запросов выполнить, и сначала просто как есть из регистра,
а потом уже с агрегированием
---
```1c
    РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, , Счет = &Займы, &ВидыСубконто, Организация = &Организация) КАК ХозрасчетныйОстаткиИОбороты
```
 вот в этой строчке убирай по одному условия и поймешь, в котором из них проблема)

```1c
    |ГДЕ
    | ХозрасчетныйОстаткиИОбороты.Субконто1 = &Контрагент
    | И ХозрасчетныйОстаткиИОбороты.Субконто2 = &Договор
```

и про эти тоже не забудь
---
проверил нормально всё выдает

```1c
    ВЫБРАТЬ
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК СуммаНачальныйОстатокДт,
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК СуммаОборотДт,
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК СуммаОборотКт,
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК СуммаКонечныйОстатокДт,
    ХозрасчетныйОстаткиИОбороты.Период КАК Период,
    ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Субконто1,
    ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Субконто2
    ИЗ
    РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, , Счет = &Займы, , Организация = &Организация) КАК ХозрасчетныйОстаткиИОбороты
    ГДЕ
    ХозрасчетныйОстаткиИОбороты.Субконто1 = &Контрагент
    И ХозрасчетныйОстаткиИОбороты.Субконто2 = &Договор

    СГРУППИРОВАТЬ ПО
    ХозрасчетныйОстаткиИОбороты.Период,
    ХозрасчетныйОстаткиИОбороты.Субконто1,
    ХозрасчетныйОстаткиИОбороты.Субконто2
```
---
```1c
    Запрос.УстановитьПараметр("Контрагент",Контрагент);
    ///////////////////
    &Контрагент
```

- тип значения я какой? План видов характеристик или просто ссылка?
---
Контрагенты.Ссылка при чем тут ПВХ?
---
Всяко разное обычно в параметры запроса суют, а потом "запрос пустой"
---
Ну вот в консоли запросов все показывает.
 Ну правда немного поменял запрос. Добавил метод дополнения "ДвиженияИГраницыПериода"

```1c
    ВЫБРАТЬ
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК СуммаНачальныйОстатокДт,
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК СуммаОборотДт,
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК СуммаОборотКт,
    СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК СуммаКонечныйОстатокДт,
    ХозрасчетныйОстаткиИОбороты.Период КАК Период,
    ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Субконто1,
    ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Субконто2
    ИЗ
    РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, ДвиженияИГраницыПериода, Счет = &Займы, &ВидыСубконто, Организация = &Организация) КАК ХозрасчетныйОстаткиИОбороты
    ГДЕ
    ХозрасчетныйОстаткиИОбороты.Субконто1 = &Контрагент
    И ХозрасчетныйОстаткиИОбороты.Субконто2 = &Договор

    СГРУППИРОВАТЬ ПО
    ХозрасчетныйОстаткиИОбороты.Период,
    ХозрасчетныйОстаткиИОбороты.Субконто1,
    ХозрасчетныйОстаткиИОбороты.Субконто2
```
---
а говорили, что не показывает
---
я вижу, например у вас на картинке результат запроса
---
Да. Но в конфигураторе тот же запрос выдает пустой результат.
---
кинь уже свой расчет процентов епэфэ...чё гадать то ))
---
причем как только комментирую отбор по договору все ок...
---
Странно , но вот так отработало
---
Т.е Ссылка на договор не равна Значения Вида Субконто Договоры...

Почему?
---
во-первых потому что отбор по владельцу должен быть
---
Что значит должен быть отбор по владельду. Так и так есть условие

ХозрасчетныйОстаткиИОбороты.Субконто1 = &Контрагент
---
ну блин ты ж про договор спрашивал - кто владелец у договора? контрагент
---
заччем там вообще виды субконто?
---
то ты пытаешься этой "обработкой" сделать?
---
Я вначале писал. Пытаюсь рассчитать проценты по договору займа. Для этого нужно получить остатки и движения по конкретному договору по дням за месяц.
---
Когда-то писал обработку по начислению процентов делал так

```1c
    Запрос = Новый Запрос;

    Запрос.УстановитьПараметр("НачалоПериода", Объект.ПериодНачисления.ДатаНачала);
    Запрос.УстановитьПараметр("КонецПериода", Объект.ПериодНачисления.ДатаОкончания);

    СписокСчетов = Новый СписокЗначений;

    //СписокСчетов.Добавить(ПланыСчетов.Хозрасчетный.ПредоставленныеЗаймы);
    //СписокСчетов.Добавить(ПланыСчетов.Хозрасчетный.НайтиПоКоду("58.03"));
    //СписокСчетов.Добавить(ПланыСчетов.Хозрасчетный.НайтиПоКоду("58.03.2"));
    СписокСчетов.Добавить(ПланыСчетов.Хозрасчетный.КраткосрочныеЗаймы);
    СписокСчетов.Добавить(ПланыСчетов.Хозрасчетный.ДолгосрочныеЗаймы);
    СписокСчетов.Добавить(ПланыСчетов.Хозрасчетный.ДолгосрочныеКредиты);

    Запрос.УстановитьПараметр("СчетаЗаймов",СписокСчетов);

    Субконто = Новый СписокЗначений;
    Субконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты);
    Субконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры);

    Запрос.УстановитьПараметр("Субконто",Субконто);

    Запрос.Текст ="ВЫБРАТЬ
    | ДанныеПроизводственногоКалендаря.Дата КАК ДатаОстатков,
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Субконто1,
    | ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Субконто2,
    | ХозрасчетныйОстаткиИОбороты.Счет КАК Счет
    |ПОМЕСТИТЬ ВТДанные
    |ИЗ
    | РегистрСведений.ДанныеПроизводственногоКалендаря КАК ДанныеПроизводственногоКалендаря,
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоПериода, &КонецПериода, , , Счет В (&СчетаЗаймов), , ) КАК ХозрасчетныйОстаткиИОбороты
    |ГДЕ
    | ДанныеПроизводственногоКалендаря.Дата МЕЖДУ &НачалоПериода И &КонецПериода
    |
    |ИНДЕКСИРОВАТЬ ПО
    | Счет,
    | ДатаОстатков
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВТДанные.ДатаОстатков КАК ДатаОстатков,
    | ВТДанные.Субконто1 КАК Субконто1,
    | ВТДанные.Субконто2 КАК Субконто2,
    | ВТДанные.Счет КАК Счет,
    | МАКСИМУМ(ХозрасчетныйОстаткиИОбороты.Период) КАК Период
    |ПОМЕСТИТЬ ВТСвернутыеДанные
    |ИЗ
    | ВТДанные КАК ВТДанные
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоПериода, &КонецПериода, День, , Счет В (&СчетаЗаймов), , ) КАК ХозрасчетныйОстаткиИОбороты
    | ПО ВТДанные.Субконто1 = ХозрасчетныйОстаткиИОбороты.Субконто1
    | И ВТДанные.Субконто2 = ХозрасчетныйОстаткиИОбороты.Субконто2
    | И ВТДанные.Счет = ХозрасчетныйОстаткиИОбороты.Счет
    | И ВТДанные.ДатаОстатков >= ХозрасчетныйОстаткиИОбороты.Период
    |
    |СГРУППИРОВАТЬ ПО
    | ВТДанные.ДатаОстатков,
    | ВТДанные.Субконто1,
    | ВТДанные.Субконто2,
    | ВТДанные.Счет
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВТСвернутыеДанные.Счет КАК Счет,
    | ВТСвернутыеДанные.Субконто1 КАК Контрагент,
    | ВТСвернутыеДанные.Субконто2 КАК Договор,
    | ЕСТЬNULL(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстаток, 0) КАК СуммаОстаток,
    | СУММА(ВЫБОР
    | КОГДА ЕСТЬNULL(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстаток, 0) = 0
    | ТОГДА 0
    | ИНАЧЕ 1
    | КОНЕЦ) КАК Дней
    |ИЗ
    | ВТСвернутыеДанные КАК ВТСвернутыеДанные
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоПериода, &КонецПериода, День, , Счет В (&СчетаЗаймов), , ) КАК ХозрасчетныйОстаткиИОбороты
    | ПО ВТСвернутыеДанные.Период = ХозрасчетныйОстаткиИОбороты.Период
    | И ВТСвернутыеДанные.Субконто1 = ХозрасчетныйОстаткиИОбороты.Субконто1
    | И ВТСвернутыеДанные.Субконто2 = ХозрасчетныйОстаткиИОбороты.Субконто2
    | И ВТСвернутыеДанные.Счет = ХозрасчетныйОстаткиИОбороты.Счет
    |ГДЕ
    | ЕСТЬNULL(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстаток, 0) <> 0
    |
    |СГРУППИРОВАТЬ ПО
    | ВТСвернутыеДанные.Счет,
    | ВТСвернутыеДанные.Субконто1,
    | ВТСвернутыеДанные.Субконто2,
    | ЕСТЬNULL(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстаток, 0)
    |
    |УПОРЯДОЧИТЬ ПО
    | Контрагент,
    | Договор,
    | Счет
    |ИТОГИ ПО
    | Контрагент,
    | Договор
    |АВТОУПОРЯДОЧИВАНИЕ" ;

    ОбходЗапроса = ОбходРезультатаЗапроса.ПоГруппировкам;

    ВыборкаКонтрагент = Запрос.Выполнить().Выбрать(ОбходЗапроса);

    ДнейВГоду = ДеньГода(КонецГода(Объект.ПериодНачисления.ДатаНачала));

    Пока ВыборкаКонтрагент.Следующий() Цикл

    ВыборкаДоговор = ВыборкаКонтрагент.Выбрать(ОбходЗапроса);

    Пока ВыборкаДоговор.Следующий() Цикл

    СтавкаИзДоговора = СтрНайти(ВыборкаДоговор.Договор.Наименование,"ЦБ РФ+1%");

    Если СтавкаИзДоговора<>0 Тогда

    ПроцентнаяСтавка = (Объект.ПроцентнаяСтавка+1)/100;

    Иначе

    ПроцентнаяСтавка = Объект.ПроцентнаяСтавка/100;

    КонецЕсли;

    СуммаПроцентов = 0;

    ВыборкаДетальныеЗаписи = ВыборкаДоговор.Выбрать();

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл

    БазаНачисления = ?(ВыборкаДетальныеЗаписи.СуммаОстаток>0,ВыборкаДетальныеЗаписи.СуммаОстаток,ВыборкаДетальныеЗаписи.СуммаОстаток*-1);

    ПроцентПоЗайму = ((БазаНачисления*ПроцентнаяСтавка)/ДнейВГоду)*ВыборкаДетальныеЗаписи.Дней;

    // СуммаПроцентов = СуммаПроцентов+((БазаНачисления*ПроцентнаяСтавка)/ДнейВГоду)*ВыборкаДетальныеЗаписи.Дней;

    СтрокаТЗ = Объект.РасчетПроцентов.Добавить();

    СтрокаТЗ.Контрагент = ВыборкаДетальныеЗаписи.Контрагент;
    СтрокаТЗ.Договор = ВыборкаДетальныеЗаписи.договор;
    СтрокаТЗ.СчетУчета = ВыборкаДетальныеЗаписи.Счет;
    СтрокаТЗ.БазаНачисления = БазаНачисления;
    СтрокаТЗ.Дней = ВыборкаДетальныеЗаписи.Дней;
    СтрокаТЗ.Процент = ПроцентПоЗайму;
    СтрокаТЗ.Ставка = ПроцентнаяСтавка*100;
    КонецЦикла;

    КонецЦикла;

    КонецЦикла;
```

Идея всего этого - получить количество дней сколько "висел" определенный остаток тела кредита т.е. если было погашения займа в середине месяца то получится N записи если было N изменений сумм тела кредита Получается такого рода таблица (например 10 числа погасили часть займа)

Контрагент | Договор | Дней | Остаток |
Контрагент 1| Договор 1| 10 | 30 000 |
Контрагент 1| Договор 1| 20 | 15 000 |
---
ТипЗначения СУБКОНТО1
Отчет производства за смену, Списание с расчетного счета, Склад, Передача ОС, Бланк строгой отчетности, Списание товаров, Авансовый отчет, ГТД по импорту, Основное средство, Корректировка поступления, Выкуп предметов лизинга, Счет-фактура выданный, Оприходование товаров, Выдача денежных документов, Передача материалов в эксплуатацию, Документ расчетов с контрагентом, Принятие к учету ОС, Поступление на расчетный счет, Корректировка реализации, Начисление пеней, Нематериальный актив, Денежный документ, Договор, Выдача наличных, Виды платежей в бюджет (фонды), Номер ГТД, Движения целевых средств, Поступление денежных документов, Оказание производственных услуг, Страна мира, Поступление НМА, Корректировка долга, Поступление (акт, накладная), Доход будущих периодов, Оказание услуг, Модернизация ОС, Прочие доходы и расходы, Назначение целевых средств, Возврат товаров от покупателя, Оплата платежной картой, Банковский счет, Отчет комиссионера (агента) о продажах, Отчет комитенту, Подразделение, Счет-фактура полученный, (не используется) Оценочные обязательства, Передача НМА, Отчет о розничных продажах, Номенклатурная группа, Уровни бюджетов, Поступление в лизинг, Объект строительства, Поступление наличных, Субконто, Поступление из переработки, Направление использования прибыли, Заявление о ввозе товаров, Контрагент, Расходы будущих периодов, Null, Реализация услуг по переработке, Комплектация номенклатуры, Реализация (акт, накладная), Виды расчетов по средствам ФСС, Виды начислений оплаты труда для налогового учета по налогу на прибыль, Возврат товаров поставщику, Ценная бумага, Отражение начисления НДС, Способы строительства, Статья затрат, Партия материалов в эксплуатации, Назначение использования, Отражение НДС к вычету, Способы учета НДС, Организация, Реализация отгруженных товаров, Ставки НДС, Виды стоимости выпуска, Регистрация в налоговом органе, Возврат материалов из эксплуатации, Начисление НДС по СМР хозспособом, (не используется) Виды страховых взносов ИП, Виды ценностей, Номенклатура, Партия, Прибыли и убытки, Оценочное обязательство, Поступление доп. расходов, Статья движения ден. средств, Виды активов и обязательств, Перемещение товаров, Физическое лицо, Элементы затрат ОписаниеТипов
---
И все равно непонятно , почему
такой запрос

```1c
    Запрос.Текст="ВЫБРАТЬ
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК СуммаНачальныйОстатокДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК СуммаОборотДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК СуммаОборотКт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК СуммаКонечныйОстатокДт,
    | ХозрасчетныйОстаткиИОбороты.Период КАК Период,
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Субконто1
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, ДвиженияИГраницыПериода, Счет = ЗНАЧЕНИЕ(ПланСчетов.Хозрасчетный.ПредоставленныеЗаймы), ЗНАЧЕНИЕ(ПланВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры), Субконто1.Наименование = &ДоговорНаименование) КАК ХозрасчетныйОстаткиИОбороты
    |
    |СГРУППИРОВАТЬ ПО
    | ХозрасчетныйОстаткиИОбороты.Период,
    | ХозрасчетныйОстаткиИОбороты.Субконто1";
```

Отрабатывает, т.е. равенство по наименованию есть.

А равенства по коду нет...

Такой запрос пустой

```1c
    Запрос.Текст="ВЫБРАТЬ
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК СуммаНачальныйОстатокДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК СуммаОборотДт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК СуммаОборотКт,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК СуммаКонечныйОстатокДт,
    | ХозрасчетныйОстаткиИОбороты.Период КАК Период,
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Субконто1
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, День, ДвиженияИГраницыПериода, Счет = ЗНАЧЕНИЕ(ПланСчетов.Хозрасчетный.ПредоставленныеЗаймы), ЗНАЧЕНИЕ(ПланВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры), Субконто1.Код = &ДоговорКод) КАК ХозрасчетныйОстаткиИОбороты
    |
    |СГРУППИРОВАТЬ ПО
    | ХозрасчетныйОстаткиИОбороты.Период,
    | ХозрасчетныйОстаткиИОбороты.Субконто1";
```
---
Как вариант - во всех справочниках, входящих в состав субконто1 есть наименование, но не во всех есть код.
Выразить(ХозрасчетныйОстаткиИОбороты.Субконто1 как Справочники.Договора) КАК Субконто1
---
Разобрался. Просто по контрагенту забито два разных договора ( с одинаковым наименованием) по разными типами. И остатки по договору с типом Прочее, а я пробую отобрать с типом "С покупателем".
---
Сейчас поясню, для чего это сделано. Чтобы правильно формировался раздел 7 декларации по НДС проценты начисляются с помощью документа Реализация, на основании него формируется счет-фактура и заполняется раздел 7 по необлагаемым операциям. А сами займы выдаются и возвращаются с помощью документов "Списание денежных средств", "Поступление денежных средств" с видом "прочие поступление", "Прочее списание".
---
И получается, что по счету 58.03 у меня договор с контрагентом с Видом договора "Прочее",
а по счету 76.03 (начисление процентов) тот же самый договор только с видом договора "С Покупателем".